



DECUS

PROGRAM LIBRARY

DECUS NO.	8-291
TITLE	TAPE TO MEMORY COMPARITOR (6-CHANNEL)
AUTHOR	T. D. Brenig-Jones
COMPANY	Digital Equipment Co., Ltd. Reading, England
DATE	July 15 1969
SOURCE LANGUAGE	PAL III

DEBOS

1911

1911

1911

TAPE TO MEMORY COMPARITOR (6-CHANNEL)

DECUS Program Library Write-up

DECUS No. 8-291

This program compares the contents of a 6-channel binary format tape with those of core memory, and prints any differences detected. Currently uses the high-speed reader, and occupies locations 7400-7577.

Start Address	7400
Error Stop	7412
End-Run Stop	7437

*7400

/TAPE-MEMORY COMPARITOR (6-CHANNEL)

START,	CLA CLL	
	RFC	/INITIALIZE H-READER
	TLS	/INITIALIZE PRINTER
	TAD RADDR	
	DCA READWD	/SET READ WORD SUBR. RETURN ADDRESS
BEGIN,	JMS READCH	/READ ONE CHAR.
	JMP .-1	/LEADER/TRAILER EXIT
	JMP READWD+4	/ORIGIN EXIT
	SNA	/DATA EXIT - BLANK?
	JMP BEGIN	/YES-IGNORE
RERROR,	HLT	/ERROR HALT
	JMP START	/RESTART
NORIG,	TAD WORD	
	DCA CLC	/RESET ORIGIN
REDATA,	JMS READWD	/READ DATA
	JMP NORIG	/ORIGIN EXIT
	TAD WORD	/DATA EXIT
CONT,	DCA HOLD	/STORE WORD JUST READ
	JMS READWD	/READ NEXT WORD
	NOP	/ORIGIN EXIT
	TAD HOLD	/DATA EXIT-FETCH LAST WORD
	CIA	/NEGATE LAST WORD
	TAD I CLC	/ADD LOCATION CONTENTS
	SZA CLA	/SAME?
	JMP CERROR	/NO-COMPARE ERROR
RETURN,	TAD ORIGIN	
	SZA CLA	/ORIGIN?
	JMP NORIG	/YES
	ISZ CLC	/UPDATE LOC'N COUNTER
	TAD WORD	/FETCH WORD JUST READ
	JMP CONT	/CONTINUE
STOPR,	HLT	/STOP RUN
	JMP START	/RESTART
READWD,	REDATA+1	/READ 12-BIT WORD SUBR.
	DCA ORIGIN	/CLEAR ORIGIN IND.
	JMS READCH	/READ ONE CHARACTER
	JMP STOPR	/LEADER/TRAILER - STOP RUN
	ISZ ORIGIN	/ORIGIN
	CLL RTL;RTL	/DATA-ROTATE 4 LEFT
	DCA WORD	/STORE 4 M.S. BITS
	JMS READCH	/READ NEXT CHARACTER
	JMP RERROR	/L/T EXIT
	JMP RERROR	/ORIGIN EXIT
	TAD WORD	/DATA EXIT
	RTL;RTL	/ROTATE 4 LEFT
	DCA WORD	/STORE 8 M.S. BITS
	JMS READCH	/READ NEXT CHARACTER
	JMP RERROR	/L/T EXIT

	JMP RERROR	/ORIGIN EXIT
	TAD WORD	/DATA EXIT
	DCA WORD	/STORE 12-BIT WORD
	SNL CLL	/ORIGIN?
	ISZ READWD	/NO-ADD 1 TO EXIT
	JMP I READWD	/EXIT SUBR.
CERROR,	TAD CR	/COMPARE ERROR
	JMS TYPE	/PRINT CR
	TAD LF	
	JMS TYPE	/PRINT LF
	TAD CLC	
	JMS OUTP	/PRINT ADDRESS
	JMS SPACES	
	TAD I CLC	
	JMS OUTP	/PRINT ADDRESS CONTENTS
	JMS SPACES	
	TAD HOLD	
	JMS OUTP	/PRINT TAPE WORD
READCH,	JMP RETURN	/RETURN TO MAIN PROGRAM
	Ø	/READ CHARACTER SUBR.
	RSF	/READER READY?
	JMP .-1	/NO-TRY AGAIN
	RFC RRB	/READ CHARACTER
	AND MASK77	/REMOVE ALL BUT 6 L.S. BITS
	DCA OUTP	/STORE CHARACTER
	TAD OUTP	
	AND MASK4Ø	
	SZA CLA	/LEADER TRAILER?
	JMP I READCH	/YES-EXIT 1
	ISZ READCH	/SET FOR EXIT 2
	TAD OUTP	
	AND MASK2Ø	
	SNA CLA	/ORIGIN?
	ISZ READCH	/NO-SET FOR EXIT 3
	TAD OUTP	
	JMP I READCH	/EXIT SUBR.
OUTP,	Ø	/PRINT OCTAL CONTENTS SUBR.
	RAL	/ALLOW FOR LINK
	DCA READCH	/HOLD WORD
	TAD MIN4	
	DCA SPACES	/SET COUNT OF 4
OUTNO,	TAD READCH	
	RTL;RAL	/ROTATE 3 LEFT
	DCA READCH	/HOLD WORD
	TAD READCH	
	AND MASK	/REMOVE ALL BUT 3 L.S. BITS
	TAD ASCOD	/CONVERT NO. TO ASCII CODE
	JMS TYPE	/PRINT ONE DIGIT
	ISZ SPACES	/ALL 4 DIGITS PRINTED?
	JMP OUTNO	/NO-CONTINUE
	JMP I OUTP	/EXIT SUBR.

SPACES,	Ø	/PRINT SPACES SUBR.
	TAD SPACE	
	JMS TYPE	/PRINT ONE SPACE
	TAD SPACE	
	JMS TYPE	/PRINT ONE SPACE
	JMP I SPACES	/EXIT SUBR.
TYPE,	Ø	/PRINT SUBR.
	TSF	/PRINTER READY?
	JMP .-1	/NO-TRY AGAIN
	TLS	/PRINT CHAR.
	CLA	
	JMP I TYPE	/EXIT SUBR.
HOLD,	Ø	
CLC,	Ø	
ORIGIN,	Ø	
WORD,	Ø	
SPACE,	24Ø	
CR,	215	
LF,	212	
MASK,	7	
MASK4Ø,	4Ø	
MASK2Ø,	2Ø	
MASK77,	77	
MIN4,	-4	
ASCOD,	26Ø	
RADDR,	REDATA+1	
\$		

*7400

/TAPE-MEMORY COMPARITOR (6-CHANNEL)

7400	7300	START,	CLA CLL	
7401	6014		RFC	/INITIALIZE H-READER
7402	6046		TLS	/INITIALIZE PRINTER
7403	1377		TAD RADDR	
7404	3241		DCA READWD	/SET READ WORD SUBR. RETURN ADDRESS
7405	4305	BEGIN,	JMS READCH	/READ ONE CHAR.
7406	5205		JMP .-1	/LEADER/TRAILER EXIT
7407	5245		JMP READWD+4	/ORIGIN EXIT
7410	7450		SNA	/DATA EXIT - BLANK?
7411	5205		JMP BEGIN	/YES-IGNORE
7412	7402	ERROR,	HLT	/ERROR HALT
7413	5200		JMP START	/RESTART
7414	1365	NORIG,	TAD WORD	
7415	3363		DCA CLC	/RESET ORIGIN
7416	4241	REDATA,	JMS READWD	/READ DATA
7417	5214		JMP NORIG	/ORIGIN EXIT
7420	1365		TAD WORD	/DATA EXIT
7421	3362	CONT,	DCA HOLD	/STORE WORD JUST READ
7422	4241		JMS READWD	/READ NEXT WORD
7423	7000		NOP	/ORIGIN EXIT
7424	1362		TAD HOLD	/DATA EXIT-FETCH LAST WORD
7425	7041		CIA	/NEGATE LAST WORD
7426	1763		TAD I CLC	/ADD LOCATION CONTENTS
7427	7640		SZA CLA	/SAME?
7430	5270		JMP CERROR	/NO-COMPARE ERROR
7431	1364	RETURN,	TAD ORIGIN	
7432	7640		SZA CLA	/ORIGIN?
7433	5214		JMP NORIG	/YES
7434	2363		ISZ CLC	/UPDATE LOC'N COUNTER
7435	1365		TAD WORD	/FETCH WORD JUST READ
7436	5221		JMP CONT	/CONTINUE
7437	7402	STOPR,	HLT	/STOP RUN
7440	5200		JMP START	/RESTART
7441	7417	READWD,	REDATA+1	/READ 12-BIT WORD SUBR.
7442	3364		DCA ORIGIN	/CLEAR ORIGIN IND.
7443	4305		JMS READCH	/READ ONE CHARACTER
7444	5237		JMP STOPR	/LEADER/TRAILER - STOP RUN
7445	2364		ISZ ORIGIN	/ORIGIN
7446	7106		CLL RTL;	
7447	7006	RTL	/DATA-ROTATE 4 LEFT	
7450	3365		DCA WORD	/STORE 4 M.S. BITS
7451	4305		JMS READCH	/READ NEXT CHARACTER
7452	5212		JMP RERROR	/L/T EXIT
7453	5212		JMP RERROR	/ORIGIN EXIT
7454	1365		TAD WORD	/DATA EXIT
7455	7006		RTL;	

7456	7006	RTL		/ROTATE 4 LEFT
7457	3365		DCA WORD	/STORE 8 M.S. BITS
7460	4305		JMS READCH	/READ NEXT CHARACTER
7461	5212		JMP RERROR	/L/T EXIT
7462	5212		JMP RERROR	/ORIGIN EXIT
7463	1365		TAD WORD	/DATA EXIT
7464	3365		DCA WORD	/STORE 12-BIT WORD
7465	7520		SNL CLL	/ORIGIN?
7466	2241		ISZ READWD	/NO-ADD 1 TO EXIT
7467	5641		JMP I READWD	/EXIT SUBR.
7470	1367	CERROR,	TAD CR	/COMPARE ERROR
7471	4354		JMS TYPE	/PRINT CR
7472	1370		TAD LF	
7473	4354		JMS TYPE	/PRINT LF
7474	1363		TAD CLC	
7475	4326		JMS OUTP	/PRINT ADDRESS
7476	4346		JMS SPACES	
7477	1763		TAD I CLC	
7500	4326		JMS OUTP	/PRINT ADDRESS CONTENTS
7501	4346		JMS SPACES	
7502	1362		TAD HOLD	
7503	4326		JMS OUTP	/PRINT TAPE WORD
7504	5231		JMP RETURN	/RETURN TO MAIN PROGRAM
7505	0000	READCH,	0	/READ CHARACTER SUBR.
7506	6011		RSF	/READER READY?
7507	5306		JMP .-1	/NO-TRY AGAIN
7510	6016		RFC RRB	/READ CHARACTER
7511	0374		AND MASK77	/REMOVE ALL BUT 6 L.S. BITS
7512	3326		DCA OUTP	/STORE CHARACTER
7513	1326		TAD OUTP	
7514	0372		AND MASK40	
7515	7640		SZA CLA	/LEADER TRAILER?
7516	5705		JMP I READCH	/YES-EXIT 1
7517	2305		ISZ READCH	/SET FOR EXIT 2
7520	1326		TAD OUTP	
7521	0373		AND MASK20	
7522	7650		SNA CLA	/ORIGIN?
7523	2305		ISZ READCH	/NO-SET FOR EXIT 3
7524	1326		TAD OUTP	
7525	5705		JMP I READCH	/EXIT SUBR.
7526	0000	OUTP,	0	/PRINT OCTAL CONTENTS SUBR.
7527	7004		RAL	/ALLOW FOR LINK
7530	3305		DCA READCH	/HOLD WORD
7531	1375		TAD MIN4	
7532	3346		DCA SPACES	/SET COUNT OF 4
7533	1305	OUTNO,	TAD READCH	
7534	7006		RTL;	

7535	7004	RAL		/ROTATE 3 LEFT
7536	3305		DCA READCH	/HOLD WORD
7537	1305		TAD READCH	
7540	0371		AND MASK	/REMOVE ALL BUT 3 L.S. BITS
7541	1376		TAD ASCOD	/CONVERT NO. TO ASCII CODE
7542	4354		JMS TYPE	/PRINT ONE DIGIT
7543	2346		ISZ SPACES	/ALL 4 DIGITS PRINTED?
7544	5333		JMP OUTNO	/NO-CONTINUE
7545	5726		JMP I OUTP	/EXIT SUBR.
7546	0000	SPACES,	0	/PRINT SPACES SUBR.
7547	1366		TAD SPACE	
7550	4354		JMS TYPE	/PRINT ONE SPACE
7551	1366		TAD SPACE	
7552	4354		JMS TYPE	/PRINT ONE SPACE
7553	5746		JMP I SPACES	/EXIT SUBR.
7554	0000	TYPE,	0	/PRINT SUBR.
7555	6041		TSF	/PRINTER READY?
7556	5355		JMP .-1	/NO-TRY AGAIN
7557	6046		TLS	/PRINT CHAR.
7560	7200		CLA	
7561	5754		JMP I TYPE	/EXIT SUBR.
7562	0000	HOLD,	0	
7563	0000	CLC,	0	
7564	0000	ORIGIN,	0	
7565	0000	WORD,	0	
7566	0240	SPACE,	240	
7567	0215	CR,	215	
7570	0212	LF,	212	
7571	0007	MASK,	7	
7572	0040	MASK40,	40	
7573	0020	MASK20,	20	
7574	0077	MASK77,	77	
7575	7774	MIN4,	-4	
7576	0260	ASCOD,	260	
7577	7417	RADDR,	REDATA+1	

